

REGION BASE

PURPOSE

Specifies the base location for regions on plots.

DESCRIPTION

On a plot, a region is defined as the figure formed by the line connecting points belonging to a common trace and a region base (typically zero). The REGION FILL command can be used to generate a solid fill or a cross-hatch fill of this region. The attributes of the fill are set with additional REGION commands (see the RELATED COMMANDS section below). The attributes of the region border are set with LINE, LINE COLOR, and LINE THICKNESS commands.

There are 3 special cases for the region base. Specifying REGION BASE POLYGON specifies that the points in the trace form a closed polygon. DATAPLOT automatically connects the first and last point if they are not already the same. This option is useful for drawing filled 2d polygons with the standard PLOT command. Drawing statistical maps is one application of this. The PROGRAM 2 example demonstrates this case.

The REGION BASE AUTOMATIC and REGION BASE INTERPOLATE commands are typically used together to fill in the area between 2 curves. The REGION BASE AUTOMATIC command specifies a variable to be used as arguments for the REGION BASE command which allows a moving region base between points. However, in the case of filling the area between curves, the base is not horizontal between points. The REGION BASE INTERPOLATE command specifies that the base should be drawn to connect pairs of base points. The PROGRAM 3 example shows how these two commands can be used to draw two curves with the area between them filled. The key trick in this macro is to define each consecutive pair of points on the second curve as a single trace. Generally, it is better practice to explicitly plot differences than it is to draw filled area curves.

The REGION BASE AUTOMATIC and REGION BASE INTERPOLATE commands are actually not needed because the REGION BASE POLYGON command supports the capability they provide in a simpler manner. The PROGRAM 4 example generates the same plot as the PROGRAM 3 example, but it uses the REGION BASE POLYGON command.

SYNTAX 1

REGION BASE <number> <number> <number> etc.

where <number> is a decimal number or parameter that specifies the desired region base. Up to 100 region bases can be specified.

SYNTAX 2

REGION BASE POLYGON <ON/OFF> <ON/OFF> etc.

where ON specifies that the trace is treated as a closed polygon while OFF specifies that it is not. Up to 100 region bases can be specified.

SYNTAX 3

REGION BASE AUTOMATIC <var>

where <var> is a variable whose elements are used as arguments to the REGION BASE command.

SYNTAX 4

REGION BASE INTERPOLATE <ON/OFF> <ON/OFF> etc.

where ON specifies that the region base is drawn between the entered values for 2 consecutive bases while OFF specifies that it is not. Up to 100 region bases can be specified.

EXAMPLES

REGION BASES 0. 0. 10.

REGION BASES 20. 20. 20.

REGION BASES 0. ALL

REGION BASES ALL 0.

REGION BASES

REGION BASE POLYGON ON

REGION BASE INTERPOLATE ON

REGION BASE AUTOMATIC Y1

NOTE 1

The REGION BASE command with no arguments sets the region base to zero for all regions. The REGION BASE command with the word ALL before or after the specified base assigns that region base to all regions; thus REGION BASE 0. ALL or REGION BASE

ALL 0. uses a base of 0. for all regions. The REGION, BAR, SPIKE, CHARACTER, and LINE switch all work independently of each other. That is, a plot point can be a line, a character, a region, a spike or a bar or any combination of the above.

NOTE 2

Traces plotted with the REGION BASE POLYGON switch set to ON do not pre-sort the data points by the horizontal axis variable (i.e., the equivalent of a PRE-SORT OFF).

DEFAULT

All region bases are 0.

SYNONYMS

None

RELATED COMMANDS

PLOT	=	Generates a data or function plot.
REGION FILL	=	Sets the on/off switches for region fills.
REGION FILL COLOR	=	Sets the color for region solid fills.
REGION PATTERN	=	Sets the types for region fill patterns.
REGION PATTERN COLOR	=	Sets the color for region hatched fills.
REGION PATTERN LINE	=	Sets the line types for region fill patterns.
REGION PATTERN SPACING	=	Sets the line spacing for region fill patterns.
REGION PATTERN THICKNESS	=	Sets the line thickness for region fill patterns.
LINE COLOR	=	Sets the color for region border lines.
LINE	=	Sets the types for region border lines.
LINE THICKNESS	=	Sets the line thickness for region border lines.

APPLICATIONS

Presentation Graphics, statistical maps, filled area charts

IMPLEMENTATION DATE


Pre-1987

The REGION BASE AUTOMATIC and REGION BASE INTERPOLATE commands and the ability to do hatch fills of non-rectangular regions were added 93/10.

The REGION BASE POLYGON option was added 94/4.


PROGRAM 1

```
LET FUNCTION F = SHIFT + AMPL*SIN(PERIOD*X+PHASE)
LET SHIFT = 3
LET AMPL = 2
LET PERIOD = 0.5
LET PHASE = PI
.
REGION FILL ON
REGION BASE SHIFT
.
TITLE SINE WAVE
TITLE SIZE 5
LET MIN = -2*PI
LET MAX = 2*PI
XLIMITS -7 7
PLOT F FOR X = MIN 0.1 MAX
```



PROGRAM 2

```
SKIP 25
READ TEXAS.DAT X Y
.
FRAME OFF
FRAME CORNER COORDINATES 5 5 95 95
MULTIPLY 2 2; MULTIPLY CORNER COORDINATES 0 0 100 100
PRE-SORT OFF
PLOT Y X
PRE-SORT ON
REGION BASE POLYGON
REGION FILL ON
PLOT Y X
REGION FILL COLOR G50
PLOT Y X
REGION PATTERN D1D2
REGION PATTERN SPACING 5
PLOT Y X
END OF MULTIPLY
```




PROGRAM 3

```

. PERCENTAGE BELOW POVERTY LEVEL
. SOURCE: "SOCIAL INDICATORS 1976", US CENSUS BUREAU
. Y1 = AGES 22-42 Y2 = AGES 65+
.
LET Y1 = DATA 9.8 9.1 8.2 7.7 8.3 8.5 8.2 7.6 7.7 8.8
LET Y2 = DATA 28.5 29.5 25.0 25.3 24.5 21.6 18.6 16.3 14.6 15.3
LET N = SIZE Y1
.
XLIMITS 1 N; XTIC OFFSET 1 1
MAJOR XTIC MARK NUMBER N; MINOR XTICMARK NUMBER 0
X1TIC MARK FORMAT ALPHA
X1TIC MARK CONTENT 66 67 68 69 70 71 72 73 74 75
YLIMITS 0 30
TITLE FILLED AREA PLOT
PLOT Y1
PRE-ERASE OFF
LET N1 = N - 1; LET N2 = N + 1
.
REGION BASE INTERPOLATE ON
REGION BASE AUTOMATIC Y1
REGION FILL ON ALL
. CREATE X AND TAG VARIABLES
LET X = SEQUENCE 1 1 N1; LET NLAST = 2*N1
LET X = SEQUENCE 2 1 N FOR I = N 1 NLAST
LET TAG = SEQUENCE 1 1 N1 FOR I = 1 1 NLAST
. CREATE Y VARIABLE
LET JUNK = Y2; DELETE JUNK(1)
DELETE Y2(^N); EXTEND Y2 JUNK
PLOT Y2 X TAG

```



PROGRAM 4

```
. PERCENTAGE BELOW POVERTY LEVEL
. SOURCE: "SOCIAL INDICATORS 1976", US CENSUS BUREAU
. Y1 = 22-42 Y2 = 65+
.
LET Y1 = DATA 9.8 9.1 8.2 7.7 8.3 8.5 8.2 7.6 7.7 8.8
LET Y2 = DATA 28.5 29.5 25.0 25.3 24.5 21.6 18.6 16.3 14.6 15.3
LET X = SEQUENCE 1966 1 1975
LET XREV = X
LET N = SIZE Y1
LET INDX = SEQUENCE N -1 1
LET INDX = SORTC INDX XREV Y2
.
EXTEND Y1 Y2
EXTEND X XREV
REGION BASE POLYGON
REGION FILL ON ALL
XLIMITS 1966 1975
XTIC OFFSET 1 1
MAJOR XTIC MARK NUMBER N
MINOR XTICMARK NUMBER 0
TITLE FILLED AREA PLOT
PLOT Y1 X
```

